Ingenieria Geologica. Curso 2006-07

LABORATORIO DE MECANICA DE MEDIOS CONTINUOS

Cuarta Sesion. Plasticidad

Problema 3

Suponiendo un modelo de plasticidad de Von Mises, obtener la condicion de plasticidad para los siguientes estados de carga:

1. tension uniaxial de traccion y de compresion σ

```
> restart;
> with(LinearAlgebra):
> contract := proc(A,B)
  local c, i, j;
  c := 0;
  for i from 1 to 3 do
  for j from 1 to 3 do
  c := c + A[i,j]*B[i,j];
  end do;
  end do;
  end proc;
```

contract := **proc**(*A*, *B*)

local *c*, *i*, *j*;

c := 0; **for** *i* **to** 3 **do for** *j* **to** 3 **do** *c* := *c* + *A*[*i*, *j*]**B*[*i*, *j*] **end do end do**

end proc

```
> mises := proc(sigma)
  local S;
  S := sigma-1/3*Trace(sigma)*IdentityMatrix(3);
  sqrt(3/2*contract(S,S));
  simplify(%,assume=positive)
  end proc;
```

mises := **proc**(σ)

local *S*;

S := σ − 1 / 3**LinearAlgebra*:-*Trace*(σ)**LinearAlgebra*:-*IdentityMatrix*(3);

sqrt(3 / 2*contract(*S*, *S*));

simplify(%, *assume* = *positive*)

end proc

```
> sigma1 := <<sigma,0,0>|<0,0,0>|<0,0,0>>;
```

$$\sigma 1 := \begin{bmatrix} \sigma & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> condicion := mises(sigma1)=sigma[f0];
```

$$condicion := \sigma = \sigma_{f0}$$

2. corte puro τ

```
> sigma2 := <<0,tau,0>|<tau,0,0>|<0,0,0>>;
```

$$\sigma2 := \begin{bmatrix} 0 & \tau & 0 \\ \tau & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> condicion := mises(sigma2)=sigma[f0];
```

$$condicion := \sqrt{3}\,\tau = \sigma_{f0}$$

3. tension uniaxial σ mas corte puro τ

```
> condicion := mises(sigma1+sigma2)=sigma[f0];
```

$$condicion := \sqrt{\sigma^2 + 3\,\tau^2} = \sigma_{f0}$$

Problema 4

Para el criterio de Mohr-Coulomb,obtener el corte de la superficie de fluencia con el plano de tension biaxial.

```
> restart;
> with(LinearAlgebra):
> mohr_coulomb :=
(sigma[1]-sigma[3])+(sigma[1]+sigma[3])*sin(phi)-2*c*cos(phi)
=0;
```

$$mohr\_coulomb := \sigma_1 - \sigma_3 + (\sigma_1 + \sigma_3)\sin(\phi) - 2\,c\cos(\phi) = 0$$

```
> st := solve(subs(sigma[3]=0,mohr_coulomb), sigma[1]);
```

$$st := \frac{2\,c\cos(\phi)}{1+\sin(\phi)}$$

```
> sc := solve(subs(sigma[1]=0,mohr_coulomb), sigma[3]);
```

$$sc := \frac{2\,c\cos(\phi)}{-1+\sin(\phi)}$$

```
> hexagono := [[st,st],[st,0],[0,sc],[sc,sc],[sc,0],[0,st]];
```

$$hexagono := \left[\left[\frac{2\,c\cos(\phi)}{1+\sin(\phi)}, \frac{2\,c\cos(\phi)}{1+\sin(\phi)}\right], \left[\frac{2\,c\cos(\phi)}{1+\sin(\phi)}, 0\right], \left[0, \frac{2\,c\cos(\phi)}{-1+\sin(\phi)}\right], \left[\frac{2\,c\cos(\phi)}{-1+\sin(\phi)}, \frac{2\,c\cos(\phi)}{-1+\sin(\phi)}\right], \left[\frac{2\,c\cos(\phi)}{-1+\sin(\phi)}, 0\right], \left[0, \frac{2\,c\cos(\phi)}{1+\sin(\phi)}\right]\right]$$

```
> hexagono := subs(c=100,phi=Pi/6,hexagono);
```

$$hexagono := \left[\left[\frac{200\cos\left(\frac{\pi}{6}\right)}{1+\sin\left(\frac{\pi}{6}\right)}, \frac{200\cos\left(\frac{\pi}{6}\right)}{1+\sin\left(\frac{\pi}{6}\right)}\right], \left[\frac{200\cos\left(\frac{\pi}{6}\right)}{1+\sin\left(\frac{\pi}{6}\right)}, 0\right], \left[0, \frac{200\cos\left(\frac{\pi}{6}\right)}{-1+\sin\left(\frac{\pi}{6}\right)}\right], \left[\frac{200\cos\left(\frac{\pi}{6}\right)}{-1+\sin\left(\frac{\pi}{6}\right)}, \frac{200\cos\left(\frac{\pi}{6}\right)}{-1+\sin\left(\frac{\pi}{6}\right)}\right], \left[\frac{200\cos\left(\frac{\pi}{6}\right)}{-1+\sin\left(\frac{\pi}{6}\right)}, 0\right], \left[0, \frac{200\cos\left(\frac{\pi}{6}\right)}{1+\sin\left(\frac{\pi}{6}\right)}\right]\right]$$

```
> plots[polygonplot](hexagono,thickness=2);
```



Problema 5

Ver enunciado entregado.

Modelo de Mohr-Coulomb

```
> restart;
> with(LinearAlgebra):
```

Meridiano de traccion

```
> sigma := <sigma_m,sigma_m,sigma_m> + <2*s/3,-s/3,-s/3>;
```

$$\sigma := \begin{bmatrix} sigma\_m + \frac{2\,s}{3} \\ sigma\_m - \frac{s}{3} \\ sigma\_m - \frac{s}{3} \end{bmatrix}$$

```
> u_oct := Normalize(<1,1,1>,Euclidean);
```

$$u\_oct := \begin{bmatrix} \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \end{bmatrix}$$

```
> ec1 := xi=simplify(u_oct.sigma);
```

$$ec1 := \xi = \sqrt{3}\ sigma\_m$$

```
> vr := simplify(sigma-(u_oct.sigma)*u_oct);
```

$$vr := \begin{bmatrix} \frac{2\,s}{3} \\ -\frac{s}{3} \\ -\frac{s}{3} \end{bmatrix}$$

```
> ec2 := rho=simplify(sqrt(vr.vr),assume=positive);
```

$$ec2 := \rho = \frac{\sqrt{6}\, s}{3}$$

```
> solve({ec1,ec2},{sigma_m,s});
```

$$\{s = \frac{\sqrt{6}\,\rho}{2},\ sigma\_m = \frac{\sqrt{3}\,\xi}{3}\}$$

```
> assign(%);
> mohr_coulomb :=
  (sigma[1]-sigma[3])+(sigma[1]+sigma[3])*sin(phi)-2*c*cos(phi)
  =0;
```

$$mohr\_coulomb := \frac{\sqrt{6}\,\rho}{2} + \left(\frac{2\sqrt{3}\,\xi}{3} + \frac{\sqrt{6}\,\rho}{6}\right)\sin(\phi) - 2\,c\cos(\phi) = 0$$

```
> rho := solve(mohr_coulomb,rho);
```

$$\rho := -\frac{2}{3}\,\frac{(\sin(\phi)\sqrt{3}\,\xi - 3\,c\cos(\phi))\sqrt{6}}{3+\sin(\phi)}$$

```
> rho_enunciado :=
  sqrt(2/3)*(2*c*cos(phi)-(2/sqrt(3))*xi*sin(phi))/(1+(1/3)*sin
  (phi));
```

$$rho\_enunciado := \frac{1}{3}\,\frac{\sqrt{6}\left(2\,c\cos(\phi) - \frac{2}{3}\sin(\phi)\sqrt{3}\,\xi\right)}{1+\frac{1}{3}\sin(\phi)}$$

```
> evalb(simplify(rho-rho_enunciado)=0);
```

true

*Resolverlo para el meridiano de compresion

```
>
```

*Modelo de Rankine

Interpretación geometrica de las superficies de fluencia

```
> restart:
> with(LinearAlgebra):
> with(plots): with(plottools):
Warning, the name changecoords has been redefined

Warning, the name arrow has been redefined
```

Dirección de la trisectriz del triedro de tensiones principales

```
> uh := Normalize(<1,1,1>,Euclidean);
```

$$uh := \begin{bmatrix} \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \\ \frac{\sqrt{3}}{3} \end{bmatrix}$$

Dirección de la proyección del eje <1,0,0> en el plano octaédrico

```
> up1 :=
  simplify(Normalize(<1,0,0>-DotProduct(<1,0,0>,uh)*uh,Euclidea
  n));
```

$$up1 := \begin{bmatrix} \frac{\sqrt{6}}{3} \\ -\frac{\sqrt{6}}{6} \\ -\frac{\sqrt{6}}{6} \end{bmatrix}$$

Dirección normal a la anterior contenida en el plano octaédrico

```
> up1a:=simplify(uh &x up1);
```

$$up1a := \begin{bmatrix} 0 \\ \frac{\sqrt{2}}{2} \\ -\frac{\sqrt{2}}{2} \end{bmatrix}$$

Vector a partir de las coordenadas locales del plano octaédrico

```
> vs := xi*uh+rho*cos(theta)*up1+rho*sin(theta)*up1a;
```

$$vs := \begin{bmatrix} \frac{\xi\sqrt{3}}{3} + \frac{1}{3}\rho\cos(\theta)\sqrt{6} \\ \frac{\xi\sqrt{3}}{3} - \frac{1}{6}\rho\cos(\theta)\sqrt{6} + \frac{1}{2}\rho\sin(\theta)\sqrt{2} \\ \frac{\xi\sqrt{3}}{3} - \frac{1}{6}\rho\cos(\theta)\sqrt{6} - \frac{1}{2}\rho\sin(\theta)\sqrt{2} \end{bmatrix}$$

Componentes globales de un cilindro de radio 1 de eje la trisectric del triedro de tensiones principales

```
> vm1 := subs(rho=1,vs[1]);
  vm2 := subs(rho=1,vs[2]);
  vm3 := subs(rho=1,vs[3]);
```

$$vm1 := \frac{\xi\sqrt{3}}{3} + \frac{1}{3}\cos(\theta)\sqrt{6}$$

$$vm2 := \frac{\xi\sqrt{3}}{3} - \frac{1}{6}\cos(\theta)\sqrt{6} + \frac{1}{2}\sin(\theta)\sqrt{2}$$

$$vm3 := \frac{\xi\sqrt{3}}{3} - \frac{1}{6}\cos(\theta)\sqrt{6} - \frac{1}{2}\sin(\theta)\sqrt{2}$$

Centro de la base del cilindro a distancia 3 del plano octaédrico

```
> basea := evalm(subs({rho=0,theta=0,xi=3},vs));
```

$$basea := [\sqrt{3}, \sqrt{3}, \sqrt{3}]$$

Cilindro de von Mises

```
> cilindro_mises :=

  plot3d([vm1,vm2,vm3],xi=-3..3,theta=0..2*Pi,axes=normal,scali
  ng=constrained,color=gold),

  spacecurve([xi,xi,xi],xi=0..3,color=black,linestyle=DASH,thic
  kness=3),

  plots[arrow](basea,uh,color=red,width=[0.15,relative]),

  plots[arrow](basea,up1,color=blue,width=[0.15,relative]),

  textplot3d({[3,0,0,‘sigma1‘],[0,3,0,‘sigma2‘],[0,0,3,‘sigma3‘
  ]},color=black):
> display([cilindro_mises]);
```



Distintas trayectorias de tensiones (uniaxial, biaxial, tracción compuesta y compresión compuesta)

```
> uniax := spacecurve([s,0,0],s=0..3,color=green,thickness=3):
  biax := spacecurve([s,s,0],s=0..3,color=blue,thickness=3):
  strac := x ->
  spacecurve([s/4,s/2,2*s/3],s=0..x,color=red,thickness=3):
  scomp := x ->
  spacecurve([-s/3,-2*s/3,-s/2],s=0..x,color=brown,thickness=3)
  :
> display(strac(5),scomp(3),axes=normal);
```




```
> sols := fsolve({s/4=vm1,s/2=vm2,2*s/3=vm3}, {s=100, theta=Pi,
  xi=3});
```

$$sols := \{s = 3.371708922, \theta = 3.550230509, \xi = 2.757764158\}$$

```
> inter_strac := sphere(subs(sols, [s/4,s/2,2*s/3]),0.2):
> sols := fsolve({-s/3=vm1,-2*s/3=vm2,-s/3=vm3}, {s=100,
  theta=0, xi=-30});
```

$$sols := \{s = 3.674234613, \theta = -1.047197551, \xi = -2.828427123\}$$

```
> inter_scomp := sphere(subs(sols, [-s/3,-2*s/3,-s/2]),0.2):
> display(cilindro_mises,uniax,biax,strac(5),scomp(5),inter_str
  ac,inter_scomp);
```



Otra forma: construcción del cilindro de von Mises en componentes de las tensiones principales

```
> sigmam:=1/3*(sigma1+sigma2+sigma3);
  ec_xi := xi=sqrt(3)*sigmam;
  ec_rho :=
  rho=sqrt((sigma1-sigmam)^2+(sigma2-sigmam)^2+(sigma3-sigmam)^
  2);
```

$$sigmam := \frac{\sigma 1}{3} + \frac{\sigma 2}{3} + \frac{\sigma 3}{3}$$

$$ec\_xi := \xi = \sqrt{3}\left(\frac{\sigma 1}{3} + \frac{\sigma 2}{3} + \frac{\sigma 3}{3}\right)$$

$$ec\_rho := \rho = \frac{\sqrt{6\,\sigma 1^2 - 6\,\sigma 1\,\sigma 2 - 6\,\sigma 1\,\sigma 3 + 6\,\sigma 2^2 - 6\,\sigma 2\,\sigma 3 + 6\,\sigma 3^2}}{3}$$

```
> implicitplot3d(rhs(ec_rho)^2=1,sigma1=0..3,sigma2=0..3,sigma3
  =0..3,grid=[20,20,20],scaling=constrained,axes=normal);
```



Superficie de fluencia de Mohr-Coulomb

```
> mohr_coulomb :=
  (sigma1-sigma3)+(sigma1+sigma3)*sin(phi)-2*c*cos(phi)=0;
```

$$mohr\_coulomb := \sigma 1 - \sigma 3 + (\sigma 1 + \sigma 3)\sin(\phi) - 2\,c\cos(\phi) = 0$$

```
> mc1 := subs(c=10,phi=Pi/6,mohr_coulomb);
```

$$mc1 := \sigma 1 - \sigma 3 + (\sigma 1 + \sigma 3)\sin\left(\frac{\pi}{6}\right) - 20\cos\left(\frac{\pi}{6}\right) = 0$$

```
> sigmam_eh:=solve(subs(sigma1=x,sigma3=x,mc1),x); # valor de x
  en la superficie de fluencia
```

$$sigmam\_eh := 10\sqrt{3}$$

```
> sigma1_u := solve(subs(sigma3=0,mc1)); # valor de sigma1 en
  la superficie de fluencia
```

$$sigma1\_u := \frac{20\sqrt{3}}{3}$$

```
> vertice := [sigmam_eh,sigmam_eh,sigmam_eh]:
> gmc1 := polygonplot3d([
  [vertice,[sigma1_u,0,0],[sigma1_u,sigma1_u,0]],
  [vertice,[sigma1_u,sigma1_u,0],[0,sigma1_u,0]],
  [vertice,[0,sigma1_u,0],[0,sigma1_u,sigma1_u]],
```

```
[vertice,[0,sigma1_u,sigma1_u],[0,0,sigma1_u]],
[vertice,[0,0,sigma1_u],[sigma1_u,0,sigma1_u]],
[vertice,[sigma1_u,0,sigma1_u],[sigma1_u,0,0]]
],axes=normal):
display(gmc1,strac(30),scomp(10));
```



Representación de la pirámide de Mohr-Coulomb con base normal a la trisectriz del triedro de tensiones principales

```
> st := solve(subs(sigma1=sm+2*s/3,sigma3=sm-s/3,sm=-10,mc1));
  sc := solve(subs(sigma1=sm+s/3,sigma3=sm-2*s/3,sm=-10,mc1));
```

$$st := \frac{60}{7} + \frac{60\sqrt{3}}{7}$$

$$sc := 12 + 12\sqrt{3}$$

```
> sm := -10;
  sigmat1 := [sm+2*st/3,sm-st/3,sm-st/3];
  sigmac1 := [sm+sc/3,sm+sc/3,sm-2*sc/3];
  sigmat2 := [sm-st/3,sm+2*st/3,sm-st/3];
  sigmac2 := [sm-2*sc/3,sm+sc/3,sm+sc/3];
  sigmat3 := [sm-st/3,sm-st/3,sm+2*st/3];
  sigmac3 := [sm+sc/3,sm-2*sc/3,sm+sc/3];
```

$$sm := -10$$

$$sigmat1 := \left[-\frac{30}{7}+\frac{40\sqrt{3}}{7}, -\frac{90}{7}-\frac{20\sqrt{3}}{7}, -\frac{90}{7}-\frac{20\sqrt{3}}{7}\right]$$

$$sigmac1 := [-6+4\sqrt{3}, -6+4\sqrt{3}, -18-8\sqrt{3}]$$

$$sigmat2 := \left[-\frac{90}{7}-\frac{20\sqrt{3}}{7}, -\frac{30}{7}+\frac{40\sqrt{3}}{7}, -\frac{90}{7}-\frac{20\sqrt{3}}{7}\right]$$

$$sigmac2 := [-18-8\sqrt{3}, -6+4\sqrt{3}, -6+4\sqrt{3}]$$

$$sigmat3 := \left[-\frac{90}{7}-\frac{20\sqrt{3}}{7}, -\frac{90}{7}-\frac{20\sqrt{3}}{7}, -\frac{30}{7}+\frac{40\sqrt{3}}{7}\right]$$

$$sigmac3 := [-6+4\sqrt{3}, -18-8\sqrt{3}, -6+4\sqrt{3}]$$

```
> baseb := [-10,-10,-10];
  uhb := [-10,-10,-10];
  upp1 := convert(simplify(10*sqrt(3)*up1),list);
```

$$baseb := [-10, -10, -10]$$

$$uhb := [-10, -10, -10]$$

$$upp1 := [10\sqrt{2}, -5\sqrt{2}, -5\sqrt{2}]$$

```
> spacecurve([x,x,x],x=-10..10*sqrt(3),color=black,thickness=3)
  ,
  plots[arrow](baseb,uhb,color=red,width=[0.15,relative]),
  plots[arrow](baseb,upp1,color=blue,width=[0.15,relative]),
  polygonplot3d([
  [vertice,sigmat1,sigmac1],
  [vertice,sigmac1,sigmat2],
  [vertice,sigmat2,sigmac2],
  [vertice,sigmac2,sigmat3],
  [vertice,sigmat3,sigmac3],
  [vertice,sigmac3,sigmat1]
  ],axes=normal),
  strac(30),scomp(30):
  display(%);
```



−30
−20
20
10
−10
−20
−30
−20

>